# $Id: PKGBUILD 84578 2013-02-20 05:00:21Z bpiotrowski $
# Maintainer:  Bartłomiej Piotrowski <nospam@bpiotrowski.pl>
# Contributor: Angel Velasquez <angvp@archlinux.org>
# Contributor: Juergen Hoetzel <juergen@archlinux.org>

pkgname=menu-cache
pkgver=0.4.1
pkgrel=1
pkgdesc='Caching mechanism for freedesktop.org compliant menus'
arch=('i686' 'x86_64')
license=('GPL2')
url='http://lxde.org/'
groups=('lxde')
depends=('intltool' 'glib2')
options=('!libtool')
makedepends=('pkgconfig')
source=(http://downloads.sourceforge.net/lxde/$pkgname-$pkgver.tar.gz)
sha256sums=('4fa9408e353fedba5b7314cbf6b6cd06d873a1424e281aa050d88bb9c0a0191e')

build() {
  cd $srcdir/$pkgname-$pkgver

  ./configure --prefix=/usr --sysconfdir=/etc \
              --libexecdir=/usr/lib/menu-cache \
              --disable-static
  make
}

package() {
  cd $srcdir/$pkgname-$pkgver
  make DESTDIR=$pkgdir install
}
